APPENDIX: LISTING OF COMPUTER PROGRAMS USED IN CHAPTER 10

Listing of FIT10.FOR

```
C  This program  FIT10.FOR  finds the best chi-square fit to the
C  yearly AIDS incidence estimates from 1979 to 1990.  This Fortran
C  program was coded by Herb Hethcote and Jim Van Ark.  It is used in
C  Chapter 10 of the book, "MODELING HIV TRANSMISSION AND AIDS IN THE
C  UNITED STATES", to obtain fits to AIDS data for homosexual men and
C  for IVDUs in aggregations of 15 subregions.  The AIDS incidence
C  data are given in vector form with names such as NE2TABT.M
C  corresponding to the notation in Table 10.1.  After the AIDS data
C  vector is entered, the parameter values are specified by the user
C  (see Tables 3.1 and 6.1).  For each parameter value, the user can
C  accept the default value or choose a new value.  Then the user
C  chooses two dates: the reduction starting year and month and the
C  reduction stopping year.  The nonlinear minimization program BCONF
C  from IMSL is used to obtain the values of PAS, RDN and ETA which
C  give the best chi-square fit to the AIDS incidence data.  For each
C  set of parameter values, BCONF calls the subroutine EPI, which
C  simulates the HIV epidemic and AIDS cases using the difference
C  equations corresponding to Figure 3.1. For each set of two dates,
C  the program displays the values of the three parameter values and
C  the chi-square value corresponding to the best fit. By trying
C  different combinations of the two dates, the user can minimize the
C  chi-square value.
C
      INTEGER N
      PARAMETER (N = 3)
      DIMENSION G(20),WRH(19),IPARAM(7),RPARAM(7),P(N),PGUESS(N),
     $ PSCALE(N),PLB(N),PUB(N),DAIDS(1975:1990)
      COMMON FH,R,QV,QA,XMU,DLT,PHI,THT,M,G,QH,WRH,NYSTRT,NMSTRT,
     $ IYEAR,IMONTH,LYEAR,NYSTOP,DAIDS
      EXTERNAL EPI,U4INF,UMACH,BCONF
      CHARACTER*10 FILE0
C
C  THE AIDS INCIDENCE FILE NAME IS ENTERED AND THE DATA IS STORED IN
C  DAIDS
      PRINT*,'INPUT A FILENAME (IN SINGLE QUOTES) FOR AIDS INCIDENCE'
      PRINT*,'BE SURE TO INCLUDE THE .m EXTENSION'
      READ*,FILE0
      DO 2 I = 1975,1978
    2 DAIDS(I) = 0.
      OPEN(UNIT=9,FILE=FILE0,STATUS='OLD')
      READ(9,*)(DAIDS(I),I=1979,1990)
          PRINT*,'DATA IN THE AIDS VECTOR STARTING IN 1975 IS',DAIDS
C
C  BASED ON DETAILED MODELING OF HOMOSEXUAL MEN IN SAN FRANCISCO, A
C  CRUDE ESTIMATE OF THE EPIDEMIC STARTING DATE IS 7 YEARS BEFORE THE
C  CUMULATIVE AIDS CASES REACH 40.  ALSO THE POPULATION SIZE CAN BE
C  CRUDELY ESTIMATED AS 7 TIMES THE CUMULATIVE AIDS INCIDENCE THROUGH
C  1990.
      FLAG = 0
      CAIDS = 0
      DO 4 I = 1975,1990
          PRCUM = CAIDS
          CAIDS = CAIDS + DAIDS(I)
        IF (FLAG.EQ.0.AND.CAIDS.GE.40) THEN
          IYEAR = I - 7
          IMONTH = INT(12*(40-PRCUM)/(CAIDS-PRCUM)+.5)
```

```
          FLAG = 1
        ENDIF
  4   CONTINUE
      NSIZE = INT(7.0*CAIDS/10000+.5)*10000
C   IO = 6 CORRESPONDS TO MONITOR OUTPUT
      IO = 6
C  STANDARD PARAMETERS WILL BE LISTED HERE.
C  LATER THE USER WILL BE ASKED IF HE/SHE WANTS TO CHANGE THEM.
      FH = .1
      R = 10
      XMU = .000532
      DLT = .05/12.
      PHI = .05/12.
      M = 7
      G(1) = .0764
      G(2) = .0665
      G(3) = .0499
      G(4) = .0429
      G(5) = .0408
      G(6) = .0529
      G(7) = .0555
      DO 5 I = 8, 20
  5   G(I) = 0
C   THE GAMMA VALUES NEED TO BE ADJUSTED SO THAT OUR DISCRETE
C   MODEL WORKS (GAMMA IS CONTINUOUS).
      DO 10  I = 1, 7
 10   G(I) = (1 - EXP(-G(I)))
      QH = .05
      WRH(1) = 2.
      WRHA = 1
      WRHP = 1.5
      WRHM = 7.5
      LYEAR = 1990
C   SETTING PARAMETERS FOR IMSL MINIMIZATION PROGRAM BCONF
      PLB(1) = .01
      PLB(2) = .01
      PLB(3) = 0.
      PUB(1) = 10
      PUB(2) = 1.5
      PUB(3) = 1.
      PSCALE(1) = 1
      PSCALE(2) = 1
      PSCALE(3) = 1
      FSCALE = 1
      ITP = 0
      RPARAM(1) = .001
      RPARAM(2) = .001
C
      PRINT*,' '
      PRINT*,'THE DEFAULT PARAMETERS VALUES ARE LISTED BELOW;'
      PRINT*,'YOU MAY CHANGE THEM OR NOT AS YOU CHOOSE.'
      PRINT*,'ANSWER 1 FOR  change  OR 2 FOR  no change.'
      PRINT*,' '
C
C  POPULATION PARAMETER VALUES
  6   PRINT*,'THE CRUDE APPROXIMATION OF ',NSIZE,' AS THE STARTING '
      PRINT*,'POPULATION SIZE IS 7.0 TIMES THE TOTAL AIDS CASES',
     $' THROUGH 1990'
      READ*,INPUT
      IF (INPUT.NE.2) THEN
```

```
            PRINT*,'INPUT NSIZE = '
            READ*,NSIZE
         ENDIF
         PRINT*,'VERY ACTIVE FRACTION FH = ',FH
         PRINT*,'DEFAULT ACTIVITY RATIO R = ',R
         READ*,INPUT
         IF (INPUT.NE.2) THEN
            PRINT*,'INPUT FH AND R'
            READ*,FH,R
         ENDIF
         SIZE = NSIZE*1.
         QV = FH*SIZE
         QA = (1-FH)*SIZE
         PRINT*, 'XMU = ',XMU,',  DLT = ',DLT,', PHI = ',PHI
         READ*,INPUT
         IF (INPUT.NE.2) THEN
            PRINT*,'XMU , DLT, PHI = '
            READ*,XMU,DLT,PHI
         ENDIF
         THT = PHI*QV/QA
C
C READ STARTING YEAR AND MONTH, THEN ENDING YEAR
         PRINT*,'A CRUDE APPROXIMATION OF THE EPIDEMIC STARTING DATE IS '
         PRINT*,'7 YEARS BEFORE THE CUMULATIVE AIDS CASES REACH 40.'
         PRINT*,'THE STARTING YEAR IS IYEAR = ',IYEAR
         PRINT*,'THE STARTING MONTH NUMBER IS IMONTH = ',IMONTH
         READ*,INPUT
         IF (INPUT.NE.2) THEN
            PRINT*,'IYEAR, IMONTH = '
            READ*,IYEAR,IMONTH
         ENDIF
         PRINT*,'THE ENDING YEAR IS LYEAR = ',LYEAR
         READ*,INPUT
         IF (INPUT.NE.2) THEN
            PRINT*,'NEW LYEAR = '
            READ*,LYEAR
         ENDIF
C
C    M IS THE NUMBER OF INFECTIOUS STAGES IN THE MODEL,
C    WHERE STAGE 1 IS THE HIV INCUBATION PHASE, STAGES 2 to INT(M/2)
C    FORM THE ASYMPTOMATIC PHASE, STAGES INT(M/2)+1 to M-1 FORM THE
C    PRE-AIDS PHASE, STAGE M IS AIDS, AND STAGE M+1 IS DEATH.
         PRINT*,'THE NUMBER OF STAGES IS M = ',M
         READ*,INPUT
         IF (INPUT.NE.2) THEN
            PRINT*,'NEW M ='
            READ*,M
         ENDIF
C
C    G IS THE VECTOR OF EXPONENTIAL WAITING TIMES gamma FOR THE STAGES
         PRINT*,'THE VALUES FOR gamma FOR THE STAGES ARE '
         PRINT*,(G(I),I=1,M)
         READ*,INPUT
         IF (INPUT.NE.2) THEN
            PRINT*,'INPUT ',M,' NEW gamma VALUES.'
            READ*,(G(I),I=1,M)
C    THE GAMMA VALUES NEED TO BE ADJUSTED SO THAT OUR DISCRETE
C    MODEL WORKS (GAMMA IS CONTINUOUS).
            DO 15  I = 1, M
 15         G(I) = (1 - EXP(-G(I)))
```

```
      ENDIF
C
C  PROBABILITY OF TRANSMISSION FOR AN ASYMPTOMATIC PERSON
C  (STAGES 2 - M/2) IS QH.
      PRINT*,'THE PROBABILITY OF TRANSMISSION BY A STAGE-2',
     $' INFECTIVE IS QH = ',QH
      READ*,INPUT
      IF (INPUT.NE.2) THEN
           PRINT*,'INPUT QH = '
           READ*,QH
      ENDIF
C
C   PROBABILITY OF TRANSMISSION IN OTHER PHASES HAS W(I) FACTOR.
      PRINT*,'SINCE THE WEIGHT FACTORS W(I) AND RELATIVE ACTIVITY',
     $' FACTORS rho APPEAR ONLY AS A PRODUCT, THEY ARE INPUT THAT',
     $' WAY.  THE PRODUCTS FOR HIV INCUBATION, ',
     $'ASYMPTOMATIC PRE-AIDS, AND AIDS PHASES ARE ',
     $ WRH(1),WRHA,WRHP,WRHM
      READ*,INPUT
      IF (INPUT.NE.2) THEN
            PRINT*,'INPUT WRH(1), WRHA, WRHP, AND WRH(M) '
            READ*,WRH(1),WRHA,WRHP,WRHM
      ENDIF
C
C  OUTPUT OF PARAMETER VALUES IN THE MODEL
      WRITE(IO,*)'PROGRAM   FIT10.FOR   '
      WRITE(IO,*)'COMPUTER SIMULATION OF HIV AND AIDS IN A POPULATION'
      WRITE(IO,*)' '
      WRITE(IO,*)'THE SIZE OF THE POPULATION IS ',NSIZE
      WRITE(IO,*)'THE VERY ACTIVE FRACTION IS ',FH
      WRITE(IO,*)'THE ACTIVITY RATIO IS ',R
      WRITE(IO,*)'THE NATURAL MORTALITY RATE XMU IS',XMU
      WRITE(IO,*)'THE TURNOVER RATE DLT IS',DLT
      WRITE(IO,*)'THE CHANGE RATE PHI FROM VERY ACTIVE TO ACTIVE',
     $' IS',PHI
      WRITE(IO,*)'THE WEIGHT FACTORS FOR INFECTIVITY ARE',
     $  WRH(1),WRHA,WRHP,WRHM
      WRITE(IO,*)'THE PROBABILITY OF TRANSMISSION IS QH = ',QH
C
C  TABLE HEADING HERE
      WRITE(IO,*)' '
      WRITE(IO,*)'          ',
     $'**************************************************'
      WRITE(IO,*)' '
      WRITE(IO,900)
  900 FORMAT(8X,'M     NYSTRT    NYSTOP    PAS    RDCTN    ETA  CHISQ')
      WRITE(IO,*) ' '
C
    8 CALL U4INF(IPARAM,RPARAM)
      PRINT*,'INPUT REDUCTION STARTING DATE, NYSTRT,NMSTRT = '
      READ*,NYSTRT,NMSTRT
      PRINT*,'THE LAST YEAR OF THE REDUCTION IS NYSTOP = '
      READ*,NYSTOP
      PRINT*,'INPUT INITIAL GUESSES FOR PAS & RDCTN '
      READ*,PGUESS(1),PGUESS(2)
      PRINT*,'INPUT INITIAL GUESS FOR ETA '
      READ*,PGUESS(3)
C
C   THE HIV INCUBATION PHASE (STAGE 1) AND THE AIDS PHASE (STAGE M)
C   HAVE FIXED GIVEN TRANSITION RATES. THE STAGES IN THE ASYMPTOMATIC
```

```
C    PHASE HAVE EQUAL LENGTH, AND THE STAGES IN THE PRE-AIDS PHASE
C    HAVE EQUAL LENGTH.
       IM2 = M/2
       WRH(M) = WRHM
       DO 20 I = 2, IM2
 20    WRH(I) = WRHA
       DO 22 I = IM2+1, M-1
 22    WRH(I) = WRHP

       CALL BCONF(EPI,N,PGUESS,ITP,PLB,PUB,PSCALE,FSCALE,IPARAM,
      $ RPARAM,P,CHISQ)
       PRINT*,'NUMBER OF FUNCTION & GRADIENT EVALUATIONS = ',
      $ IPARAM(3),IPARAM(5)

 950   FORMAT(1X,3I5,4F10.4)
       WRITE(IO,950)NYSTRT,NMSTRT,NYSTOP,P(1),P(2),P(3),CHISQ
       PRINT*,'DO YOU WISH TO DO IT AGAIN?  (2 = yes)'
       READ*,INPUT
       IF (INPUT.EQ.2) THEN
           PRINT*,'DO YOU WISH ONLY TO CHANGE  nystrt,nmstrt,nystop,',
      $            'pas,rdctn & eta?'
           READ*,INPUT
           IF (INPUT.EQ.2) GOTO 8
           GOTO 6
       ENDIF
       END
C
C  ***************************************************************
C
C    SUBROUTINE TO SIMULATE HIV EPIDEMIC
       SUBROUTINE EPI(N,P,CHISQ)
       DIMENSION X(20),Y(20),Z(20),DX(20),DY(20),DZ(20),G(20),
      $ WRH(19),DAIDS(1975:1990),P(N)
       COMMON FH,R,QV,QA,XMU,DLT,PHI,THT,M,G,QH,WRH,NYSTRT,NMSTRT,
      $ IYEAR,IMONTH,LYEAR,NYSTOP,DAIDS
C
       PAS = P(1)
       RDCTN = P(2)
       ETA = P(3)
       PHS = PAS/(1 + FH*(R-1))
       RDCTMO = RDCTN**(1./12.)
C
C  INITIALIZING ON IMONTH OF IYEAR
       START = 1
       DO 50 I = 1,M+1
       X(I) = 0
       Y(I) = 0
  50   Z(I) = 0
       X(1) = START
       SV = QV - X(1)
       SA = QA
C
C  STARTING THE ITERATION, WITH STEP SIZE = 1 MONTH
       NMONTH = IMONTH
       LASTK = LYEAR - IYEAR
       PRAIDS = 0
       CHISQ = 0
       DO  600  K = 0,  LASTK
       NYEAR = IYEAR + K
C
```

```
C    J  IS THE NUMBER OF THE MONTH
       DO  500 J = NMONTH, 12
C
C  ACTIVE HOMOSEXUALS HAVE PH PARTNERS PER MONTH
C  VERY ACTIVE HOMOSEXUALS HAVE R*PH PARTNERS PER MONTH
       IF (NYEAR.LT.NYSTRT) THEN
             PH = PHS
          ELSEIF (NYEAR.EQ.NYSTRT.AND.J.LT.NMSTRT) THEN
             PH = PHS
          ELSEIF (NYEAR.LE.NYSTOP) THEN
             PH = PHS*RDCTMO**((NYEAR-NYSTRT-1)*12+13-NMSTRT+J)
          ELSE
             PH = PHS*RDCTMO**((NYSTOP-NYSTRT)*12+13-NMSTRT)
       ENDIF
C
C  CALCULATING THE INCIDENCES
       ASUM = 0
       VSUM = 0
       DO 100 I=1,M
       ASUM = ASUM + WRH(I)*Y(I)
 100   VSUM = VSUM + WRH(I)*R*X(I)
       SUM = ASUM + VSUM
       DNM = QA - Y(M+1) + R*(QV - X(M+1))
       AINC = PH*QH*((1.- ETA)*ASUM*SA/(QA - Y(M+1)) + ETA*SUM*SA/DNM)
       VINC = PH*QH*((1.-ETA)*VSUM*SV/(QV-X(M+1)) + ETA*SUM*R*SV/DNM)
C
C  CALCULATING THE DIFFERENCES
       DSV = (DLT+XMU)*(QV-SV)-VINC-PHI*SV+THT*SA
       DSA = (DLT+XMU)*(QA-SA)-AINC+PHI*SV-THT*SA
       DX(1) = VINC-(G(1)+XMU+DLT+PHI)*X(1)+THT*Y(1)
       DY(1) = AINC - (G(1)+XMU+DLT+THT)*Y(1)+PHI*X(1)
       DZ(1) = DLT*(X(1)+Y(1))-(G(1)+XMU)*Z(1)
       DO 200 I=2,M
       DX(I) = G(I-1)*X(I-1)+THT*Y(I)-(G(I)+XMU+PHI+DLT)*X(I)
       DY(I) = G(I-1)*Y(I-1)+PHI*X(I)-(G(I)+XMU+THT+DLT)*Y(I)
 200   DZ(I) = G(I-1)*Z(I-1)+DLT*(X(I)+Y(I))-(G(I)+XMU)*Z(I)
       DZ(M) = DZ(M) - G(M-1)*Z(M-1)
       ZAIDS = ZAIDS + G(M-1)*Z(M-1)
       DX(M+1) = G(M)*X(M)+THT*Y(M+1)-(XMU+PHI+DLT)*X(M+1)
       DY(M+1) = G(M)*Y(M)+PHI*X(M+1)-(XMU+THT+DLT)*Y(M+1)
       DZ(M+1) = (DLT+XMU)*(X(M+1)+Y(M+1)) + G(M)*Z(M)
       DZ(M+1) = DZ(M+1) + XMU*(X(M) + Y(M) + Z(M))
C
C  UPDATING EACH CLASS
       SV = SV + DSV
       SA = SA + DSA
       DO 300 I=1,M+1
       X(I) = X(I) + DX(I)
       Y(I) = Y(I) + DY(I)
 300   Z(I) = Z(I) + DZ(I)
C
C  CHECKING CONSERVATION IN QV AND QA
       CKQV = SV
       CKQA = SA
       DO 400 I = 1, M+1
       CKQV = CKQV + X(I)
 400   CKQA = CKQA + Y(I)
       IF (ABS(CKQV-QV)/QV + ABS(CKQA-QA)/QA.GT.1.E-4) THEN
            PRINT*,'CONSERVATION CHECK NOT SATISFIED'
            PRINT*,QV,CKQV,QA,CKQA
```

```
      ENDIF
 500  CONTINUE
C
C  YEARLY OUTPUT
      DEATHS = X(M+1) + Y(M+1) + Z(M+1)
      AIDS = DEATHS + X(M) + Y(M) + Z(M)
      AIDINC = AIDS - PRAIDS
      IF(NYEAR.GE.1978.AND.NYEAR.LE.1990)THEN
            CHISQ = CHISQ + (DAIDS(NYEAR)-AIDINC)**2 / AIDINC
      ENDIF
      PRAIDS = AIDS
 600  NMONTH = 1
      RETURN
      END
```

Listing of IVDU10.FOR

```
C  This program  IVDU10.FOR  is used for fitting to heterosexual and
C  perinatal AIDS data and also for producing tables of simulation
C  values.  This program was coded by Herb Hethcote and Jim Van Ark.
C  It is used in Chapter 10 of the book,  "MODELING HIV TRANSMISSION
C  AND AIDS IN THE UNITED STATES",  to do fitting and to produce
C  Tables 10.5, 10.6, 10.8, 10.9, 10.12 and 10.15. The program
C  HOMO10.FOR for producing Tables 10.2, 10.3, 10.4, 10.7, 10.10,
C  10.11, 10.13 and 10.14 for homosexual men is similar, but simplier
C  since there are no heterosexual partners. For an aggregated
C  population, the AIDS incidence data from 1979 to 1990 for IVDUs,
C  heterosexual cases and perinatal cases are input to the program as
C  vectors such as NE2TABT.M, NE5TABT.M AND NECTABT.M.  After the
C  parameter values found by FIT10.FOR for the best fit to the AIDS
C  data for IVDUs are entered into this program, the user runs the
C  program in terminal mode to find the value of the parameter PAP,
C  the average number of new heterosexual partners per month, which
C  gives the best fit.  This PAP value is found by a user-directed
C  iterative search to minimize the chi-square value of the fit to the
C  heterosexual AIDS data.  After the value of PAP has been found, the
C  fecundity FC is found by a user-directed iterative search to
C  minimize the chi-square value of the fit to the perinatal AIDS
C  data.  When the best fitting values of PAP and FC have been found,
C  then the printer mode is used to produce a table of parameter
C  values followed by simulation values for the HIV epidemic and AIDS
C  cases in IVDUs (I), their heterosexual partners (P), and perinatal
C  cases in children (C).  The program also produces external files of
C  the bimonthly HIV and AIDS incidence values for making MATLAB
C  graphs from 1974 to 1990 using the program GRIVDU10.M.
C
      DIMENSION X(20),Y(20),Z(20),DX(20),DY(20),DZ(20),GA(20),GP(20),
     $ WRH(19),DIVDU(1975:1990),DHTRO(1975:1990),DPED(1975:1990),
     $ NAIVDU(102),NHIVDU(102),NAHTRO(102),NHHTRO(102),NAPED(102),
     $ NHPED(102),YP(20),ZP(20),DYP(20),DZP(20),YC(20),DYC(20)
C
      CHARACTER*10 FILE0I,FILE1I,FILE2I,FILE0H,FILE1H,FILE2H,
     $ FILE0C,FILE1C,FILE2C
C
C  THE AIDS INCIDENCE FILE NAMES ARE ENTERED AND THE DATA ARE STORED
C  IN VECTORS
      PRINT*,'INPUT A FILENAME (IN SINGLE QUOTES) FOR IVDU INCIDENCE'
      PRINT*,'BE SURE TO INCLUDE THE .m EXTENSION'
```

```
      READ*,FILE0I
C  NOTE:  INPUT FILE  ALWAYS STARTS IN 1979!
      INYEAR = 1979
      DO 5 N = 1975,INYEAR-1
      DIVDU(N) = 0.
      DHTRO(N) = 0.
   5  DPED(N) = 0.
      OPEN(UNIT=9,FILE=FILE0I,STATUS='OLD')
      READ(9,*)(DIVDU(N),N=INYEAR,1990)
      PRINT*,'DATA IN IVDU VECTOR STARTING IN 1975 IS',DIVDU
      PRINT*,'INPUT THE FRACTION OF IVDUs WHO ARE WOMEN'
      READ*,PIW
C
      PRINT*,'INPUT THE FILENAME (SINGLE QUOTES) FOR HTRO INCIDENCE'
      PRINT*,'BE SURE TO INCLUDE THE .m EXTENSION'
      READ*,FILE0H
      OPEN(UNIT=9,FILE=FILE0H,STATUS='OLD')
      READ(9,*)(DHTRO(N),N=INYEAR,1990)
          PRINT*,'DATA IN HTRO VECTOR STARTING IN 1975 IS',DHTRO
      PRINT*,'INPUT THE FRACTION OF HETEROSEXUALS WHO ARE WOMEN'
      READ*,PHW
C
      PRINT*,'INPUT THE FILENAME (IN SINGLE QUOTES) FOR PED INCIDENCE'
      PRINT*,'BE SURE TO INCLUDE THE .m EXTENSION'
      READ*,FILE0C
      OPEN(UNIT=9,FILE=FILE0C,STATUS='OLD')
      READ(9,*)(DPED(N),N=INYEAR,1990)
          PRINT*,'DATA IN PED VECTOR STARTING IN 1975 IS',DPED
C
C    THE DEFAULT NAMES FOR THE EXTERNAL DATA FILES TO BE WRITTEN TO
C    ARE GIVEN HERE; ONE CAN CHANGE THESE NAMES AT THE TIME OF WRITING.
          FILE1I = 'MTGRF1.M'
          FILE2I = 'MTGRF2.M'
          FILE1H = 'MTGRF3.M'
          FILE2H = 'MTGRF4.M'
          FILE1C = 'MTGRF5.M'
          FILE2C = 'MTGRF6.M'
      DATA NAIVDU /102*0/, NHIVDU /102*0/
      DATA NAHTRO /102*0/, NHHTRO /102*0/
      DATA NAPED  /102*0/, NHPED  /102*0/
C
C  PARAMETERS WILL BE LISTED HERE.  LATER USER WILL BE ASKED ABOUT
C  CHANGING.
      NSIZE = 150000
      NSIZEP = NSIZE/2
      FH = .1
      R = 10
      XMU = .000532
      DLT = .05/12.
      PHI = .05/12.
      ETA = .5
      M = 7
      GA(1) = .0764
      GA(2) = .0665
      GA(3) = .0499
      GA(4) = .0429
      GA(5) = .0408
      GA(6) = .0529
      GA(7) = .0555
      DO 7 I = 8, 20
```

```
 7     GA(I) = 0
       G1 = .4571
       G2 = .0190
       G3 = .0159
       GM = .0555
C
C  THE GAMMA VALUES NEED TO BE ADJUSTED SO THAT OUR DISCRETE MODEL
C  WORKS (GAMMA IS CONTINUOUS)
       DO 9  I=1,M
  9    GA(I)=(1-EXP(-GA(I)))
       QH = .05
       QHP = .1
       QC = .1
       WRH(1) = 2.
       WRHA = 1.
       WRHP = 1.5
       WRHM = 7.5
       NYSTRT = 1981
       NMSTRT = 1
       NYSTOP = 1986
       PAS = .7972
       RDCTN = .6889
       PAP = .0181
       FC = .0012
       A = 1.55
       PC = .34
       B = .08
       IYEAR = 1974
       IMONTH = 1
       LYEAR = 1990
       START = 1
 11    PRINT*,'INPUT 6 FOR TERMINAL OUTPUT AND 8 FOR PRINTER OUTPUT'
       READ*,IO
       PRINT*,' '
       PRINT*,'THE DEFAULT STARTING PARAMETERS WILL BE LISTED;'
       PRINT*,'YOU MAY CHANGE THEM OR NOT AS YOU CHOOSE.'
       PRINT*,'ANSWER 1 FOR  change  OR 2 FOR  no change.'
       PRINT*,' '
C
C  POPULATION PARAMETER VALUES
       PRINT*,'THE IVDU & HTRO POPULATION SIZES ARE ',NSIZE,NSIZEP
       READ*,INPUT
       IF (INPUT.NE.2) THEN
            PRINT*,'INPUT NSIZE & NSIZEP = '
            READ*,NSIZE,NSIZEP
       ENDIF
       PRINT*,'VERY ACTIVE FRACTION FH = ',FH
       PRINT*,'DEFAULT ACTIVITY RATIO R = ',R
       READ*,INPUT
       IF (INPUT.NE.2) THEN
            PRINT*,'INPUT FH AND R'
            READ*,FH,R
       ENDIF
       SIZE = NSIZE*1.
       QV = FH*SIZE
       QA = (1-FH)*SIZE
       QAP = NSIZEP*1.
       PRINT*, 'XMU = ',XMU,',  DLT = ',DLT,', PHI = ',PHI
       READ*,INPUT
       IF (INPUT.NE.2) THEN
```

```
            PRINT*,'XMU , DLT, PHI = '
            READ*,XMU,DLT,PHI
        ENDIF
        THT = PHI*QV/QA
C
C    M IS THE NUMBER OF INFECTIOUS STAGES IN THE MODEL, WHERE
C    STAGE 1 IS THE HIV INCUBATION PHASE, STAGES 2 to INT(M/2) FORM THE
C    ASYMPTOMATIC PHASE, STAGES INT(M/2)+1 to M-1 FORM THE PRE-AIDS
C    PHASE, STAGE M IS AIDS, AND STAGE M+1 IS DEATH.
        PRINT*,'THE NUMBER OF INFECTIOUS STAGES IS  M = ',M
        READ*,INPUT
        IF (INPUT.NE.2) THEN
             PRINT*,'INPUT NUMBER OF STAGES, M = '
             READ*,M
        ENDIF
C
C    ADJUSTMENT FOR gamma's FOR PEDIATRIC CASES.
C    THE HIV INCUBATION PHASE (STAGE 1) AND THE AIDS PHASE (STAGE M)
C    HAVE FIXED GIVEN TRANSITION RATES.  THE  STAGES IN THE
C    ASYMPTOMATIC PHASE HAVE EQUAL LENGTH, AND THE STAGES IN THE PRE-
C    AIDS PHASE HAVE EQUAL LENGTH. GP IS FOR PEDIATRIC gamma's.
        GP(1) = G1
        IM2=INT(M/2)
        DO 18 I = 2, IM2
 18     GP(I) = G2*(IM2-1)
        DO 19 I=IM2+1, M-1
 19     GP(I) = G3*(M-1-IM2)
        GP(M) = GM
C
C   THE GAMMA VALUES NEED TO BE ADJUSTED SO THAT OUR DISCRETE MODEL
C   WORKS (GAMMA IS CONTINUOUS)
        DO 21  I = 1, M
 21     GP(I)=(1-EXP(-GP(I)))
C
C    NOW CHECK THE ADULT gamma's.
        PRINT*,'THE gamma''s FOR ADULTS ARE ',(GA(I), I = 1, M)
        READ*,INPUT
        IF (INPUT.NE.2) THEN
            PRINT*,'INPUT ',M,' NEW gamma VALUES.'
            READ*,(GA(I), I = 1, M)
            DO 23  I = 1, M
 23         GA(I)=(1-EXP(-GA(I)))
        ENDIF
C
C    PROBABILITIES OF TRANSMISSION FOR ASYMPTOMATICS ARE QH FOR
C    NEEDLE-SHARING, QHP FOR HETEROSEXUAL PARTNERS AND QC FOR CHILDREN
        PRINT*,'THE PROBABILITIES OF TRANSMISSION BY ASYMPTOMATIC',
       $' INFECTIVES ARE QH, QHP, QC = ',QH,QHP,QC
        READ*,INPUT
        IF (INPUT.NE.2) THEN
             PRINT*,'INPUT QH, QHP & QC = '
             READ*,QH,QHP,QC
        ENDIF
        PRINT*,'SINCE THE WEIGHT FACTORS W(I) AND RELATIVE ACTIVITY',
       $' FACTORS rho APPEAR ONLY AS A PRODUCT, THEY ARE INPUT THAT',
       $' WAY.  THE PRODUCTS FOR HIV INCUBATION, ',
       $'ASYMPTOMATIC PRE-AIDS, AND AIDS PHASES ARE ',
       $ WRH(1),WRHA,WRHP,WRHM
        READ*,INPUT
        IF (INPUT.NE.2) THEN
```

```
            PRINT*,'INPUT WRH(1), WRHA, WRHP, AND WRH(M) '
            READ*,WRH(1),WRHA,WRHP,WRHM
      ENDIF
      DO 20 I = 2, IM2
 20   WRH(I) = WRHA
      DO 22 I = IM2+1,M-1
 22   WRH(I) = WRHP
      WRH(M) = WRHM
C
C   MIXING BETWEEN ACTIVITY LEVELS IN PREFERRED MIXING WITH ETA AS
C   THE EXTERNAL FRACTION, AND (1 - ETA) AS THE INTERNAL FRACTION
      PRINT*,'THE EXTERNAL MIXING FRACTION IS ETA = ',ETA
      READ*,INPUT
      IF (INPUT.NE.2) THEN
            PRINT*,'THE EXTERNAL MIXING FRACTION ETA ='
            READ*,ETA
      ENDIF
C
C READ STARTING YEAR AND MONTH, THEN ENDING YEAR
      PRINT*,'THE STARTING YEAR IS IYEAR = ',IYEAR
      PRINT*,'THE STARTING MONTH NUMBER IS IMONTH = ',IMONTH
      READ*,INPUT
      IF (INPUT.NE.2) THEN
            PRINT*,'IYEAR, IMONTH = '
            READ*,IYEAR,IMONTH
      ENDIF
      PRINT*,'THE ENDING YEAR IS LYEAR = ',LYEAR
      READ*,INPUT
      IF (INPUT.NE.2) THEN
             PRINT*,'NEW LYEAR = '
             READ*,LYEAR
      ENDIF
C
C  THE INITIAL NUMBER OF INFECTIVES IN THE POPULATION IS START.
C  THEY WILL ALL BE PUT IN THE VERY ACTIVE CLASS EVENTUALLY.
      PRINT*,'THE INITIAL NUMBER OF INFECTIVES,  START = ',START
      READ*,INPUT
      IF (INPUT.NE.2) THEN
            PRINT*,'START = '
            READ*,START
      ENDIF
C
C  NEEDLE-SHARING BEHAVIOR PARAMETERS (RDCTN = YEARLY REDUCTION
C  STARTING IN NYSTRT, NMSTRT AND ENDING IN NYSTOP.)
      PRINT*,'THE STARTING TIME FOR REDUCTION(NYSTRT,NMSTRT) = ',
     $ NYSTRT,NMSTRT
      READ*,INPUT
      IF (INPUT.NE.2) THEN
            PRINT*,'INPUT NYSTRT,NMSTRT = '
            READ*,NYSTRT,NMSTRT
      ENDIF
      PRINT*,'THE LAST YEAR FOR THE REDUCTION IS NYSTOP = ', NYSTOP
      READ*,INPUT
      IF (INPUT.NE.2) THEN
            PRINT*,'INPUT NYSTOP = '
            READ*,NYSTOP
      ENDIF
 32   PRINT*,'THE AVERAGE NUMBER OF NEEDLE-SHARING PARTNERS PER',
     $' MONTH BEFORE ',NYSTRT,NMSTRT,' IS PAS = ',PAS,' AND THE',
     $' DEFAULT YEARLY REDUCTION IS RDCTN = ',RDCTN,' UNTIL DEC, ',
```

```
     $NYSTOP
      READ*,INPUT
      IF (INPUT.NE.2) THEN
            PRINT*,'NEW PAS, RDCTN =  '
            READ*,PAS,RDCTN
      ENDIF
      PHS = PAS/(1+FH*(R-1))
      RDCTMO = RDCTN**(1./12.)
      PRINT*,'THE AVERAGE NUMBER OF NEW IVDU PARTNERS PER MONTH ',
     $'IS ',PAP
      READ*,INPUT
      IF (INPUT.NE.2) THEN
            PRINT*,'NEW PAP = '
            READ*,PAP
      ENDIF
C  THE PROPORTION PC OF HIV CHILDREN DEVELOP AIDS EARLY FROM YCE TO
C  Y(M) WITH RATE CONSTANT B.  THE OTHERS MOVE MORE RAPIDLY THROUGH
C  THE M STAGES WITH G(I) MULTIPLIED BY A.
      PRINT*,'FOR 1-PC CHILDREN, gamma''s ARE MULTIPLIED BY A = ',A
      READ*,INPUT
      IF (INPUT.NE.2) THEN
            PRINT*,'NEW A ='
            READ*,A
      ENDIF
      PRINT*,'EARLY AIDS FRACTION PC & RATE CONSTANT B = ', PC,B
      READ*,INPUT
      IF (INPUT.NE.2) THEN
            PRINT*,'NEW PC,B ='
            READ*,PC,B
      ENDIF
C  THE FECUNDITY OF IVDU & HETEROSEXUAL WOMEN IS FC
      PRINT*,'THE FECUNDITY IS FC = ',FC
      READ*,INPUT
      IF (INPUT.NE.2) THEN
            PRINT*,'NEW FC ='
            READ*,FC
      ENDIF
C
      NMONTH = IMONTH
C
C  INITIALIZING ON IMONTH OF IYEAR
      DO 50 I = 1,M+1
      X(I) = 0
      Y(I) = 0
      Z(I) = 0
      YP(I) = 0
      ZP(I) = 0
   50 YC(I) = 0
      YCE = 0
      X(1) = START
      SV = QV - X(1)
      SA = QA
      SAP = QAP
C
C  OUTPUT OF PARAMETER VALUES IN THE MODEL
      WRITE(IO,*)'TABLE'
      WRITE(IO,*)' '
      WRITE(IO,*)FILEOI,FILEOI,FILEOI,FILEOI,FILEOI
      WRITE(IO,*)'*************************************************'
      WRITE(IO,*)'PROGRAM OUTPUT FOR IVDU10.FOR  BY H. W. HETHCOTE'
```

```
      WRITE(IO,*)'COMPUTER SIMULATION OF HIV AND AIDS IN AN',
     $' IVDU POPULATION '
      WRITE(IO,*)' '
      WRITE(IO,*)'THE IVDU & HTRO POPULATION SIZES ARE ',NSIZE,NSIZEP
      WRITE(IO,*)'THE VERY ACTIVE FRACTION IS ',FH
      WRITE(IO,*)'THE ACTIVITY RATIO IS ',R
      WRITE(IO,*)'THE NATURAL MORTALITY RATE XMU IS',XMU
      WRITE(IO,*)'THE INTERCHANGE RATE  FROM THE VERY ACTIVE',
     $' CLASS TO THE ACTIVE CLASS IS'
      WRITE(IO,*) PHI,'  AND THE TURNOVER RATE IS DLT = ',DLT
      WRITE(IO,*)'THE NUMBER OF INFECTIOUS STAGES IS  M = ',M
      WRITE(IO,*)'THE G PARAMETERS FOR THE TRANSFER BETWEEN',
     $' ADULT STAGES ARE ',(GA(I), I = 1, M)
      WRITE(IO,*)'THE WEIGHTS OF TRANSMISSION PER INFECTIOUS ',
     $'PARTNER TIMES THE FRACTION STILL'
      WRITE(IO,*)'SEXUALLY ACTIVE FOR THE STAGES ARE WRH(I) = ',
     $ (WRH(I), I = 1, M)
      WRITE(IO,*)'THE PROBABILITIES OF TRANSMISSION ARE QH, QHP &',
     $' QC = ',QH,QHP,QC
      WRITE(IO,*)'THE EXTERNAL MIXING FRACTION IS ETA = ',ETA
      WRITE(IO,*)'THE AVERAGE NUMBER OF NEEDLE-SHARING PARTNERS PER ',
     $'MONTH IS ',PAS,'     BEFORE ',NYSTRT,NMSTRT,', THEN IT IS ',
     $'REDUCED EACH YEAR BY A FACTOR OF ',RDCTN,' UNTIL DEC, ',NYSTOP
      WRITE(IO,*)'THE FRACTION OF IVDU WHO ARE WOMEN IS ',PIW
      WRITE(IO,*)'THE FRACTION OF HETEROSEXUALS WHO ARE WOMEN IS ',PHW
      WRITE(IO,*)'THE AVERAGE NUMBER OF IVDU PARTNERS OF HETERO',
     $'SEXUALS PER MONTH IS',PAP
      WRITE(IO,*)'THE FRACTION',PC,' OF CHILDREN PROGRESS RAPIDLY TO',
     $' AIDS WITH RATE   CONSTANT',B,'. OTHERS MOVE THROUGH M STAGES',
     $' WITH SPEED FACTOR',A
      WRITE(IO,*)'THE FECUNDITY FC (CHILDREN/MONTH) IS',FC
      WRITE(IO,*)'THE STARTING YEAR AND MONTH ARE ',IYEAR,IMONTH
      WRITE(IO,*)'THE STARTING NUMBER OF VERY ACTIVE INFECTIVES IS',
     $ START
C
C  TABLE HEADING HERE
      WRITE(IO,*)'*************************************************'
      WRITE(IO,*)' '
      WRITE(IO,*)'THE SIMULATED INCIDENCES ARE GIVEN ON THE NEXT PAGE'
      WRITE(IO,850)
 850  FORMAT('1')
      WRITE(IO,*)FILE0I,FILE0I,FILE0I,FILE0I,FILE0I
      WRITE(IO,*)'*************************************************'
      WRITE(IO,900)
 900  FORMAT(1X,'YEAR          HIV     HIV  FRACTNAL_PREV ',
     $'  YR AIDS INC   AIDS(SIMULATION)')
      WRITE(IO,901)
 901  FORMAT(1X,'     CLASS    INC    PREV  ALL  V_A  ACT ',
     $'   DATA   SIM   PREV   DTHS  OUTSF')
      WRITE(IO,*) ' '
C
C  STARTING THE ITERATION, WITH STEP SIZE = 1 MONTH
      LASTK = LYEAR - IYEAR
      PRAIDS = 0
      PRDTHS = 0
      PRZAID = 0
      ZAIDS = 0
      CHISQ = 0
      PRAIDP = 0
      PRDTHP = 0
```

```
      PRZAIP = 0
      ZAIDSP = 0
      CHISQP = 0
      PRAIDC = 0
      PRDTHC = 0
      CHISQC = 0
      DO 600 K=0,LASTK
      SUMAIN = 0
      SUMVIN = 0
      SUMAIP = 0
      SUMAIC = 0
      NYEAR = IYEAR + K
C
C  J IS THE NUMBER OF THE MONTH
      DO 500 J=NMONTH,12
C
C  ACTIVE IVDU'S HAVE PH PARTNERS PER MONTH
C  VERY ACTIVE IVDU'S HAVE R*PH PARTNERS PER MONTH
      IF (NYEAR.LT.NYSTRT) THEN
            PH = PHS
         ELSEIF (NYEAR.EQ.NYSTRT.AND.J.LT.NMSTRT) THEN
            PH = PHS
         ELSEIF (NYEAR.LE.NYSTOP) THEN
            PH = PHS*RDCTMO**((NYEAR-NYSTRT-1)*12+13-NMSTRT+J)
         ELSE
            PH = PHS*RDCTMO**((NYSTOP-NYSTRT)*12+13-NMSTRT)
      ENDIF
C
C  CALCULATING THE MONTHLY INCIDENCES
      ASUM = 0
      VSUM = 0
      SUMP = 0
      SUMC = 0
      DO 100 I=1,M
      ASUM = ASUM + WRH(I)*Y(I)
      VSUM = VSUM + WRH(I)*R*X(I)
      SUMP = SUMP + WRH(I)*(X(I)+Y(I))
 100  SUMC = SUMC + WRH(I)*YP(I)
      SUM = ASUM + VSUM
      DNM = QA - Y(M+1) + R*(QV - X(M+1))
      AINC = PH*QH*((1.-ETA)*ASUM*SA/(QA-Y(M+1))+ETA*SUM*SA/DNM)
      VINC = PH*QH*((1.-ETA)*VSUM*SV/(QV-X(M+1))+ETA*SUM*R*SV/DNM)
      AINCP = PAP*QHP*SUMP*SAP/(QAP-YP(M+1))
C  PIW OF IVDU & PHW OF HETEROSEXUALS ARE WOMEN
      AINCC = FC*QC*(PIW*SUMP+PHW*SUMC)
      SUMAIN = SUMAIN + AINC
      SUMVIN = SUMVIN + VINC
      SUMAIP = SUMAIP + AINCP
      SUMAIC = SUMAIC + AINCC
C
C  FIRST LOAD THE VECTORS FOR GRAPHING !
C  {NOTICE THAT THIS ROUTINE WILL LOAD DATA ON THE EVEN MONTHS.}
      IF (NYEAR.GE.1974.AND.NYEAR.LE.1990) THEN
        IF(INT(J/2)*2.EQ.J) THEN
         N4 = (NYEAR - 1974) * 6 + INT(J/2)
         NAIVDU(N4)=GA(M-1)*(X(M-1)+Y(M-1))
         NHIVDU(N4) = AINC + VINC
         NAHTRO(N4)=GA(M-1)*YP(M-1)
         NHHTRO(N4) = AINCP
         NAPED(N4)=A*GP(M-1)*YC(M-1)+B*YCE
```

```
          NHPED(N4) = AINCC
        ENDIF
      ENDIF
C
C  CALCULATING THE DIFFERENCES
      DSV = (DLT+XMU)*(QV-SV)-VINC-PHI*SV+THT*SA
      DSA = (DLT+XMU)*(QA-SA)-AINC+PHI*SV-THT*SA
      DSAP = (DLT+XMU)*(QAP-SAP)-AINCP
      DX(1) = VINC-(GA(1)+XMU+DLT+PHI)*X(1)+THT*Y(1)
      DY(1) = AINC - (GA(1)+XMU+DLT+THT)*Y(1)+PHI*X(1)
      DZ(1) = DLT*(X(1)+Y(1))-(GA(1)+XMU)*Z(1)
      DYP(1) = AINCP - (GA(1)+XMU+DLT)*YP(1)
      DZP(1) = DLT*YP(1)-(GA(1)+XMU)*ZP(1)
      DYC(1) = (1-PC)*AINCC - A*GP(1)*YC(1)
      DO 200 I=2,M
      DX(I) = GA(I-1)*X(I-1)+THT*Y(I)-(GA(I)+XMU+PHI+DLT)*X(I)
      DY(I) = GA(I-1)*Y(I-1)+PHI*X(I)-(GA(I)+XMU+THT+DLT)*Y(I)
      DZ(I) = GA(I-1)*Z(I-1)+DLT*(X(I)+Y(I))-(GA(I)+XMU)*Z(I)
      DYP(I) = GA(I-1)*YP(I-1)-(GA(I)+XMU+DLT)*YP(I)
      DZP(I) = GA(I-1)*ZP(I-1)+DLT*YP(I)-(GA(I)+XMU)*ZP(I)
  200 DYC(I) = A*GP(I-1)*YC(I-1) - A*GP(I)*YC(I)
      DZ(M) = DZ(M) - GA(M-1)*Z(M-1)
      ZAIDS = ZAIDS + GA(M-1)*Z(M-1)
      DZP(M) = DZP(M) -GA(M-1)*ZP(M-1)
      ZAIDSP = ZAIDSP +GA(M-1)*ZP(M-1)
      DX(M+1) = GA(M)*X(M)+THT*Y(M+1)-(XMU+PHI+DLT)*X(M+1)
      DY(M+1) = GA(M)*Y(M)+PHI*X(M+1)-(XMU+THT+DLT)*Y(M+1)
      DZ(M+1) = (DLT+XMU)*(X(M+1)+Y(M+1)) + GA(M)*Z(M)
      DZ(M+1) = DZ(M+1) + XMU*(X(M) + Y(M) + Z(M))
      DYP(M+1) = GA(M)*YP(M)-(XMU+DLT)*YP(M+1)
      DZP(M+1) = (DLT+XMU)*YP(M+1) + GA(M)*ZP(M) + XMU*(YP(M)+ZP(M))
      DYC(M) = DYC(M) + B*YCE
      DYC(M+1) = A*GP(M)*YC(M)
C
C  UPDATING EACH CLASS
      SV = SV + DSV
      SA = SA + DSA
      SAP = SAP + DSAP
      DO 300 I=1,M+1
      X(I) = X(I) + DX(I)
      Y(I) = Y(I) + DY(I)
      Z(I) = Z(I) + DZ(I)
      YP(I) = YP(I) + DYP(I)
      ZP(I) = ZP(I) + DZP(I)
  300 YC(I) = YC(I) + DYC(I)
      YCE = YCE + PC*AINCC - B*YCE
C
C  CHECKING CONSERVATION IN QV AND QA
      CKQV = SV
      CKQA = SA
      CKQAP = SAP
      DO 400 I = 1, M+1
      CKQV = CKQV + X(I)
      CKQA = CKQA + Y(I)
  400 CKQAP = CKQAP + YP(I)
      IF (ABS(CKQV-QV)/QV + ABS(CKQA-QA)/QA.GT.1.E-5) THEN
          PRINT*,'CONSERVATION CHECK NOT SATISFIED (IVDU)'
          PRINT*,QV,CKQV,QA,CKQA
      ENDIF
      IF (ABS(CKQAP-QAP)/QAP.GT.1.E-5) THEN
```

```
          PRINT*,'CONSERVATION CHECK NOT SATISFIED (HETERO)'
          PRINT*,QAP,CKQAP
        ENDIF
  500   CONTINUE
C
C  YEARLY OUTPUT
        SUMINC =  SUMAIN + SUMVIN
        SUMINP = SUMAIP
        XPREV = 0
        YPREV = 0
        ZPREV = 0
        YPREVP = 0
        ZPREVP = 0
        PREVC = YCE
        DO 550 I = 1, M
        XPREV = XPREV + X(I)
        YPREV = YPREV + Y(I)
        ZPREV = ZPREV + Z(I)
        YPREVP = YPREVP + YP(I)
        ZPREVP = ZPREVP + ZP(I)
  550   PREVC = PREVC + YC(I)
        PREV = XPREV + YPREV
        IF (QV.GT.0) FXPREV = XPREV / QV
        IF (QA.GT.0) FYPREV = YPREV / QA
        FPREV = PREV / NSIZE
        PREVP = YPREVP
        FPPREV = YPREVP / NSIZEP
        DEATHS = X(M+1) + Y(M+1) + Z(M+1)
        AIDPRV =  X(M) + Y(M) + Z(M)
        AIDS = DEATHS + AIDPRV
        YRDTHS = DEATHS - PRDTHS
        AIDINC = AIDS - PRAIDS
        OUTINC = ZAIDS - PRZAID
        DEATHP = YP(M+1) + ZP(M+1)
        AIDPRVH =  YP(M) + ZP(M)
        AIDSP = DEATHP + AIDPRVH
        YRDTHP = DEATHP - PRDTHP
        AIDINP = AIDSP - PRAIDP
        OUTINP = ZAIDSP - PRZAIP
        DEATHC = YC(M+1)
        AIDPRVC = YC(M)
        AIDSC = DEATHC + AIDPRVC
        YRDTHC = DEATHC - PRDTHC
        AIDNCC = AIDSC - PRAIDC
  910   FORMAT(1X,I4,1X,A4,2F8.0,3F5.2,F7.0,5F7.0)
  911   FORMAT(1X,I4,1X,A4,2F8.0,3F5.2,'   ****',5F7.0)
  912   FORMAT(7X,A4,2F8.0,F5.2,'  -    -  ',5F7.0)
  913   FORMAT(7X,A4,2F8.0,F5.2,'  -    -     ****',5F7.0)
  914   FORMAT(7X,A4,2F8.0,'  -    -    -  ',5F7.0)
  915   FORMAT(7X,A4,2F8.0,'  -    -    -     ****',5F7.0)
        IF (NYEAR.GE.1975.AND.NYEAR.LE.1990.AND.AIDINC.GE.1.E-5) THEN
              CHISQ = CHISQ + (DIVDU(NYEAR)-AIDINC)**2 / AIDINC
             WRITE(IO,910) NYEAR,'IVDU',SUMINC,PREV,FPREV,FXPREV,FYPREV,
     $                      DIVDU(NYEAR),AIDINC,AIDPRV,YRDTHS,OUTINC
           ELSE
            WRITE(IO,911) NYEAR,'IVDU',SUMINC,PREV,FPREV,FXPREV,FYPREV,
     $                      AIDINC,AIDPRV,YRDTHS,OUTINC
        ENDIF
        IF (NYEAR.GE.1975.AND.NYEAR.LE.1990.AND.AIDINP.GE.1.E-5) THEN
              CHISQP = CHISQP + (DHTRO(NYEAR)-AIDINP)**2 / AIDINP
```

```
          WRITE(IO,912) 'HTRO',SUMINP,PREVP,FPPREV,DHTRO(NYEAR),
     $                  AIDINP,AIDPRVH,YRDTHP,OUTINP
        ELSE
          WRITE(IO,913) 'HTRO',SUMINP,PREVP,FPPREV,AIDINP,AIDPRVH,
     $                   YRDTHP,OUTINP
        ENDIF
        IF (NYEAR.GE.1975.AND.NYEAR.LE.1990.AND.AIDNCC.GE.1.E-5) THEN
          CHISQC = CHISQC + (DPED(NYEAR)-AIDNCC)**2 / AIDNCC
          WRITE(IO,914) 'PED',SUMAIC,PREVC,DPED(NYEAR),
     $                  AIDPRVC,AIDNCC,YRDTHC
        ELSE
          WRITE(IO,915) 'PED',SUMAIC,PREVC,AIDNCC,AIDPRVC,YRDTHC
        ENDIF

        PRDTHS = DEATHS
        PRAIDS = AIDS
        PRZAID = ZAIDS
        PRDTHP = DEATHP
        PRAIDP = AIDSP
        PRZAIP = ZAIDSP
        PRDTHC = DEATHC
        PRAIDC = AIDSC
600     NMONTH = 1
        WRITE(IO,*)' '
        WRITE(IO,*)'CHISQD = ',CHISQ
        WRITE(IO,*)'CHISQP = ',CHISQP
        WRITE(IO,*)'CHISQC = ',CHISQC
        CHISQT = CHISQ + CHISQP + CHISQC
        WRITE(IO,*)'SUM OF CHISQ-D,P,C =',CHISQT
C
C   CHECK TO SEE IF THE DATA IS TO BE USED FOR GRAPHING.
        PRINT*,'WOULD YOU LIKE TO WRITE DATA FILES TO EXTERNAL '
     $'(GRAPHING) FILES?  2 = YES.'
        READ*,INPUT
        IF (INPUT.EQ.2) THEN
        PRINT*,'THE DEFAULT NAMES FOR IVDU FILES ARE ',FILE1I,FILE2I
            READ*,INPUT
            IF (INPUT.NE.2) THEN
               PRINT*,'INPUT NEW DATA FILE NAMES (IN SINGLE QUOTES)'
               READ*,FILE1I,FILE2I
            ENDIF
            OPEN(UNIT=10,FILE=FILE1I)
            WRITE(10,*)NHIVDU
            OPEN(UNIT=10,FILE=FILE2I)
            WRITE(10,*)NAIVDU
C
        PRINT*,'THE DEFAULT NAMES FOR HTRO FILES ARE ',FILE1H,FILE2H
            READ*,INPUT
            IF (INPUT.NE.2) THEN
               PRINT*,'INPUT NEW DATA FILE NAMES (IN SINGLE QUOTES)'
               READ*,FILE1H,FILE2H
            ENDIF
            OPEN(UNIT=10,FILE=FILE1H)
            WRITE(10,*)NHHTRO
            OPEN(UNIT=10,FILE=FILE2H)
            WRITE(10,*)NAHTRO
C
        PRINT*,'THE DEFAULT NAMES FOR THE PED FILES ARE ',FILE1C,FILE2C
            READ*,INPUT
            IF (INPUT.NE.2) THEN
```

```
            PRINT*,'INPUT THE NEW DATA FILE NAMES (IN SINGLE QUOTES)'
            READ*,FILE1C,FILE2C
          ENDIF
          OPEN(UNIT=10,FILE=FILE1C)
          WRITE(10,*)NHPED
          OPEN(UNIT=10,FILE=FILE2C)
          WRITE(10,*)NAPED
        ENDIF
C
        PRINT*,'INPUT 1 TO STOP, OR 2 TO DO IT AGAIN'
        READ*,INPUT
        IF (INPUT.NE.2) GOTO 999
        DO 700 I=1,102
        NHIVDU(I)=0
        NAIVDU(I)=0
        NHHTRO(I)=0
        NAHTRO(I)=0
        NHPED(I)=0
 700    NAPED(I)=0
        GO TO 11
 999    END
```

Listing of GRIVDU10.M

```
%   The program  GRIVDU10.M  is a MATLAB graphing program using pre-
%   generated vectors for the estimated and simulated HIV and AIDS
%   incidence from 1974 to 1990.  The AIDS incidence data vector names
%   (such as NE2TABT, NE5TABT, NECTABT) are entered first.   Then the
%   simulation vectors for HIV and AIDS pre-generated by the Fortran
%   program IVDU10.FOR with default names mtgrf#.m  (# =1 to 6) are
%   entered.  This program then produces the graphs of the AIDS
%   incidence data and the HIV and AIDS simulation incidences for the
%   IVDUs, heterosexual partners and perinatally-infected children.
%   This program was coded by Herb Hethcote and Jim Van Ark.  It is
%   used in Chapter 10 of the book, "MODELING HIV TRANSMISSION AND
%   AIDS IN THE UNITED STATES", to produce Figures 10.18, 10.19,
%   10.26, 10.27, 10.43 and 10.54.
%
   n = 3
%
%  Input the names of the vectors for inputing data and labeling the
%  graphs.
    m1 = input('Input the vector name for IVDUs (in single-quotes) ');
    l1 = [m1,' solid'];
           m2 = input('Input the vector name for HTRO ');
           l2 = [m2,' dashed'];
               m3 = input('Input the vector name for PED  ');
               l3 = [m3,' dashdot'];
%
%  Y0, Y0H, Y0C are yearly AIDS data for IVDU, HTRO & PED from  CDC
   file0 = m1;
   eval(['load ',file0,'.m'])
   eval(['Y0 = ',file0,';'])
   Y0 = Y0*(1/12);
   Y0 = Y0(:);
   file0H = m2;
   eval(['load ',file0H,'.m'])
```

```
eval(['Y0H = ',file0H,';'])
Y0H = Y0H*(1/12);
Y0H = Y0H(:);
file0C = m3;
eval(['load ',file0C,'.m'])
eval(['Y0C = ',file0C,';'])
Y0C = Y0C*(1/12);
Y0C = Y0C(:);
%
%   These matrices contain the x - entries for the data points
X0 = 1979+1/2:1:1990+1/2;
X = 1974:1/6:1991 - 1/12;
%
%   Load the simulation vectors for IVDU
disp('Use 2 to accept or 1 to change.    ');
inp = input('the default names for IVDU are mtgrf1 & mtgrf2  ');
if inp == 1
   file1 = input('Input file1 name (in single quotes)  ');
   file2 = input('Input file2 name  ');
else
   file1 = 'mtgrf1';
   file2 = 'mtgrf2';
end;
eval(['load ',file1,'.m']);
eval(['load ',file2,'.m']);
eval(['Y1 = ',file2,';'])
eval(['Z1 = ',file1,';'])
Y1 = Y1';
Z1 = Z1';
Y1 = Y1(:);
Z1 = Z1(:);
%
%   Load the simulation vectors for HTRO
    inp = input('the defaults for HTRO are mtgrf3 & mtgrf4  ');
          if inp == 1
             file3 = input('input file3 name  ');
             file4 = input('Input file4 name  ');
          else
             file3 = 'mtgrf3';
             file4 = 'mtgrf4';
          end;
          eval(['load ',file3,'.m']);
          eval(['load ',file4,'.m']);
          eval(['Y2 = ',file4,';'])
          eval(['Z2 = ',file3,';'])
          Y2 = Y2';
          Z2 = Z2';
          Y2 = Y2(:);
          Z2 = Z2(:);
%
%   Load the simulation vectors for PED
    inp = input('the defaults for PED are mtgrf5 & mtgrf6  ');
          if inp == 1
             file5 = input('input file5 name  ');
             file6 = input('Input file6 name  ');
          else
             file5 = 'mtgrf5';
             file6 = 'mtgrf6';
          end;
          eval(['load ',file5,'.m']);
```

```
            eval(['load ',file6,'.m']);
            eval(['Y3 = ',file6,';'])
            eval(['Z3 = ',file5,';'])
            Y3 = Y3';
            Z3 = Z3';
            Y3 = Y3(:);
            Z3 = Z3(:);
%
   pack
%
%    Plotting
  plot(X0,Y0,':',X0,Y0,'x',X0,Y0H,':',X0,Y0H,'x',X0,Y0C,':',X0,...
   Y0C,'x',X,Y1,'-',X,Y2,'--',X,Y3,'-.',X,Z1,'-',X,Z2,'--',X,Z3,'-.')
          title('MONTHLY INCIDENCES')
          xlabel('Year')
          ylabel('Incidence')
          text(.2,.82,l1,'sc')
          text(.2,.76,l2,'sc')
          text(.2,.7,l3,'sc')
          text(.6,.8,'HIV','sc')
          text(.7,.4,'AIDS','sc')
          pause
          ni = input('do you wish to print this?(1 = yes)  ');
          if ni == 1
             print
          end;
     end
```